

パソコンを使ったグラフのかき方 gnuplot 編*

※二次配布禁止※

「減衰振動の測定」担当TA： 奈須田 祐大[†]

Ver. 1.2.2

目 次

1 導入	1
1.1 gnuplot とは	1
1.2 インストール方法	2
2 gunplot を用いたグラフのかき方・fitting の方法 ①	2
2.1 データの整理	2
2.2 グラフのかき方	3
2.3 fitting の方法 ①（線形の場合）	9
2.4 完成形	10
3 gunplot を用いた fitting の方法 ②	10
3.1 fitting の方法 ②（非線形 fitting）	10
4 おわりに	12

1 導入

1.1 gnuplot とは

フリーのグラフ描画ソフト。「gnuplot」は、「ニュープロット」などと発音される。様々な関数のグラフをかいたり、データをプロットしたりすることが、(初めはコマンドの入力に抵抗を覚えるかもしれないけど) 簡単に、しかも“きれいに”できる。更に、数値計算の結果などの大量のデータをプロットするのに向いていたり、物理屋の必需品である(?) I^AT_EX との相性が良かったりするために、広く用いられている。

*筆者が「gnuplot 編」の必要性を感じたのは、第一に、筆者の研究分野及びその周辺では論文に MS Excel でかかれたグラフを載せる例は見たことがないこと、第二に、慣れ親しんだものにしか手を出さず新しいものを敬遠する態度（謂わば“MS-Office 鎖国”？）はあまりにも消極的過ぎるということ、が挙げられる。

[†]Email: 6221702@ed.tus.ac.jp

1.2 インストール方法

gnuplot のホームページ (<http://www.gnuplot.info>) などからインストールできる。参考までに、2020 年 10 月現在のバージョンは、5.4.0 である。ただし、以下の説明ではバージョン 5.2.8 を用いる（おそらく大差はない筈）。また、筆者が使っている OS は、Windows 10（の Pro）である。

詳しくは、google などの検索エンジンで検索すれば分かるので、ここでは割愛する（どの OS にインストールするかによって、多少操作が異なる場合があるので注意¹⁾）。

2 gunplot を用いたグラフのかき方・fitting の方法 ①

この章では、物理学実験 1-A の第 3 講「最小二乗法・グラフの書き方」でグラフのかき方や最小二乗法の計算練習に用いたデータ「金属試料の長さとの関係」を例に、gnuplot の使い方を説明してゆく。つまり、既に学んだことと全く同じことを“手”ではなく gnuplot でやる方法を説明する、ということだ（何か新しいことを学ぶ際に、既に知っていることをその新しい方法でもできるか、ということを確認めるのは、理解を確認める上で非常に重要なことである）。

2.1 データの整理

gnuplot でデータをプロットするには、まず、「メモ帳」などのエディタで以下のような dat ファイル（データファイルの一種；拡張子は .dat）を作る。ただし、gnuplot においては、「#」はコメント（コンピュータが処理しない行）を表す。

```
# data.dat
#
# Temperature [degree Celsius]      Length [m]
10      1.00021
20      1.00036
30      1.00053
40      1.00074
50      1.00091
60      1.00106
```

保存する際、拡張子を .dat にする。例えば、Windows の「メモ帳」なら、保存する際に「ファイル形式」を「すべて」にして、ファイル名の末尾に .dat と書く。ファイル名は何でもよいが、日本語（全角文字）を使うべきではない。また、このファイルが保存されている場所を覚えておくこと。尚、保存場所は、例えば Windows なら、「プロパティ」から調べることができる。以下の説明では、「data.dat」が「C:\Users\admin\Documents\gnuplot」²⁾ に保存されているとする。

¹⁾ macOS なら、note の記事 <https://note.com/nppp/n/n92dc11fdfecc> の「おまけ」などを参照。

²⁾ 環境によっては、バックスラッシュ「\」ではなく、円記号「¥」になる。

2.2 グラフのかき方

2.2.1 gnuplot の起動

gnuplot を起動すると、以下の画面が現れる。

```
G N U P L O T
Version 5.2 patchlevel 8    last modified 2019-12-01

Copyright (C) 1986-1993, 1998, 2004, 2007-2019
Thomas Williams, Colin Kelly and many others

gnuplot home:      http://www.gnuplot.info
faq, bugs, etc:    type "help FAQ"
immediate help:    type "help"  (plot window: hint 'h')

Terminal type is now 'wxt'
gnuplot> _
```

まず初めに、作業ディレクトリを、前節で作った dat ファイルが保存されているフォルダ「C:\Users\admin\Documents\gnuplot」に変更する。コマンド「cd」は、change directory の意味である。

```
cd 'C:\Users\admin\Documents\gnuplot'
```

また、後のために、文字コードを UTF-8 に変更しておく。

```
set encoding utf8
```

gnuplot を終了するとき

「閉じるボタン」をクリックするのではなく、

```
exit
```

または

```
quit
```


と入力することで終了させる。

2.2.2 データのプロット

次に、前節で作った dat ファイル「data.dat」を読み込んで、そのデータをグラフ化する方法を述べる。

```
plot "data.dat" using 1:2
```

とすれば、「data.dat」というファイルの1列目を横軸に、2列目を縦軸にとって、プロットすることができる。これを実行すると、次のようなグラフを得る。



参考 関数のグラフ

ここでは、関数のグラフのかき方を説明する。gnuplot で使える二項演算子や組込関数は、適宜調べること。


関数のグラフをかく場合にも、次のように「plot」というコマンドを使えばよい。例えば、関数 $f(x) = \sin x$ をかくなら、

```
plot sin(x)
```


とする。少し角ばっているのが気になるときは、

```
set samples 2000
```

などとすればよい（数字が大きい程、曲線は滑らか）。



(a) sample 数 50 のとき.



(b) sample 数 2000 のとき.

2.2.3 グラフの修正

前小節で、データをプロットしたグラフをかくことができた。しかしながら、上の方法でかいたグラフには、以下の問題点がある。

- グラフタイトルがない
- 軸ラベルがない
- データ点と枠が重なっていて見づらい
- 目盛りの数字の位が揃っていない
- 副目盛りがない
- 凡例がジャマ（データ点との区別が付きづらい）
- プロット（+印）が見づらい

以下では、これらの解決方法を示してゆく。

■ グラフタイトルの書き方

gnuplot で「set title」というコマンドを使う手もあるが、 $\text{\LaTeX}$ で文書を書いていたたり Beamer でスライドを作成していたりする場合には figure 環境の「caption」で、MS PowerPoint でスライドを作成してる場合は「テキストボックス」で、グラフタイトルを書くのが一般的である。

■ 軸ラベルの付け方

軸ラベルを「Temperature [°C]」「Length [m]」のように付けるには、

```
set xlabel 'Temperature [{\Symbol \260}C]'  
set ylabel 'Length [m]'
```

のように入力する。「set xlabel」は横軸のラベルを設定するコマンドで、シングル・クォテーション内の文字列が出力される。ただし、「{\Symbol \260}」は、「°」を出力するためのコマンドである。このように、特別な記号については、直接入力できずコマンドで入力しなければならない場合がある。また、「set ylabel」は縦軸のラベルを設定するコマンドである。

軸ラベルを「Temperature, $t/^{\circ}\text{C}$ 」「Length, l/m 」のようにしたい場合には、

```
set xlabel 'Temperature, {\Arial-Italic t}/{\Symbol \260}C'  
set ylabel 'Length, {\Arial-Italic l}/m'
```

とする。{\Arial-Italic t}は、 t を斜体にする³⁾、というコマンドである。

更に、軸ラベルのフォントサイズを大きくしたいときは、

```
set xlabel '{/=30 Temperature [{\Symbol \260}C]}'  
set ylabel '{/=30 Length [m]}'
```

などとする（デフォルトは/=10）。

■ グラフ範囲の変更の仕方

グラフの範囲（横軸の範囲と縦軸の範囲）を指定するには、「set xrange」「set yrange」というコマンドが使える。

```
set xrange [0:70]  
set yrange [1:1.0012]
```

と入力すれば、横軸の最小値と最大値がそれぞれ0と70、縦軸の最小値と最大値はそれぞれ1と1.0012へと変更される。

³⁾ 厳密には、物理量は「斜体」ではなく「イタリック体」で書くべきである。例えば、「 x 」は「 x 」ではなく「 x 」と書く。

■ 目盛りの書式設定の仕方

まず、目盛りの数字の位を揃えるためには、以下のように桁数を指定すればよい。

```
set format x "%2.0f"  
set format y "%5.4f"
```

この例では、横軸は2桁で小数点以下はなし、縦軸は5桁でそのうち4桁が小数点以下、となるようになっている。

フォントサイズを大きくしたければ、

```
set tics font "Arial,30"
```

のように入力する（デフォルトは Arial,10）。

更に、横軸の主目盛りの間隔を20に、縦軸の主目盛りの間隔を0.0002にした上で、副目盛りは縦横共に主目盛りを2等分するように付けるためには、

```
set xtics 20  
set ytics 0.0002  
set mxtics 2  
set mytics 2
```

とすればよい。

■ 凡例の調整の仕方

今回の例のように、グラフに示すデータが1種類しかない場合、凡例を示す必要はない。この場合、

```
unset key
```

と入力すれば、凡例が消える。

ただし、凡例を示したい場合には、

```
set key box
```

で凡例を枠で囲んで、データ点との区別を明確にするとよい。更に、

```
set key left top
```

のようにして、凡例の位置を変えることも可能（この例だと、グラフ中左上に凡例が表示される）。また、凡例を表示する場合には、その名前も適切に変える必要がある。データ点を（「plot」のコマンドを使って）プロットする際に、

```
plot "data.dat" using 1:2 title "data"
```

のようになると、ダブル・クォーテーション内の文字列が凡例に表示される。

■ プロット点の変更

プロット点を、デフォルトの「+」から「●」に変更するには、データ点を（「plot」のコマンドを使って）プロットする際に、

```
plot "data.dat" using 1:2 pt 7 lc 8
```

と入力する。「pt」はpoint typeでプロット点の種類を、「lc」はline colorでプロットの色を表している。「7」や「8」の数字は、それぞれプロット点の種類やプロットの色を具体的に指定している。プロット点の種類やプロットの色は、これら以外にも様々あるので、工夫して使い分けるとよい（どの形の点やどの色が、どの数字で指定されるかは、適宜調べること）。

2.2.4 保存方法

gnuplotでは、描画したグラフを、様々なファイル形式で出力することができる。例えば、直前にかいたグラフを「graph1.png」という名前のpngファイルとして出力・保存したければ、

```
set terminal png
set output "graph1.png"; replot
set terminal wxt
```

とすればよい。ここで、「replot」は、直前にかいたものと同一のものを出力するためのコマンドである。また他にも、pdf形式で出力・保存したければ、「png」となっている部分を「pdf」に、eps形式で出力・保存したければ、「png」となっている部分を「eps」にすればよい。

尚、出力した画像ファイルは、現在の作業ディレクトリである「C:\Users\admin\Documents\gnuplot」に保存される。

2.3 fittingの方法 ① (線形の場合)

2.3.0 fittingとは

得られたデータに“最もよく合う = fit する”ような直線または曲線（しばしば関数のパラメータ）を求めること。得られた直線や曲線は、それぞれ近似直線や近似曲線とよばれる。近似直線や近似曲線を表す関数は、最小二乗法によって求められる。

2.3.1 fittingの方法

gnuplot で fitting を行う際には、まず初めに、fit させる関数形を指定する。ここでは、fit させる関数は、1 次関数 $ax + b$ で、これを「f(x)」とおくことにする。Fitting をするためのコマンドは「fit」で、

```
f(x) = a*x + b
fit f(x) "data.dat" using 1:2 via a,b
```

のように入力する。この 2 行目は、関数「f(x)」に「data.dat」ファイルのデータを、1 列目を変数 x 、2 列目を $f(x)$ の値、 a, b をパラメータとして fitting する、という意味である。これを実行すると、最小二乗法の計算が始まり、計算が終わると、

Final set of parameters	Asymptotic Standard Error
=====	=====
a = 1.74571e-05	+/- 4.199e-07 (2.405%)
b = 1.00002	+/- 1.635e-05 (0.001635%)

という計算結果が表示される。従って、

$$a = (1.75 \pm 0.04) \times 10^{-5}, \quad b = 1.00002 \pm 0.00002$$

と求まる。即ち、求める近似直線の式は、

$$l/m = (1.75 \pm 0.04) \times 10^{-5} t / ^\circ\text{C} + 1.00002 \pm 0.00002$$

である。

尚、この fitting の記録（ログ）は、現在の作業ディレクトリ内に「fit.log」というファイルが生成され、そこに残されている。

近似直線をデータ点と共にプロットしたグラフをかくには、

```
plot "data.dat" using 1:2 pt 7 lc 8, f(x) lc 8
```

のように、プロットしたいものをコンマで繋げばよい。

2.4 完成形

前節までに説明した方法でかいて完成したグラフを、図1に示す。




図 1: 金属試料の長さとの温度の関係

3 gunplot を用いた fitting の方法 ②

“複雑な”関数に fit させたいとき、前章の方法では、(拡張された最小二乗法の) 計算が収束しなくなってしまう。本章では、この場合の対処法を、以下に示す過制動の測定結果を例に説明する。

3.1 fitting の方法 ② (非線形 fitting)

3.1.1 使用するデータ

本章で使用するデータは、「減衰振動の測定」におけるグリセリン水溶液中で円筒の運動が過制動となるときの測定結果である。そのデータを以下に示す（グラフ化する際は、1列目と3列目のみを用いる）。

```
# gly4.dat
#
# t[s]      x[mm]      theta[rad]
0          300      0.15053
5           95      0.07141
```

10	-30	0.02316
15	-70	0.00772
20	-86	0.00154
25	-88	0.00077
30	-90	0.00000

3.1.2 非線形 fitting の方法

“複雑な”関数に fit させるときには，fitting する前に「初期値」を設定する必要がある．パラメータのおおよその値の見当を付け，その値を「初期値」として与える．この「初期値」の与え方が悪いと，これもまた計算が収束しなくなってしまうため，値の見当の付ける際には注意が必要だ．データと関数形とから，適当な初期値を見出す必要がある．

今回の例では，

$$Ae^{\frac{-k+\sqrt{k^2-4Ic}}{2I}x} + Be^{\frac{-k-\sqrt{k^2-4Ic}}{2I}x}$$

という形の関数に，前小節に示したデータを fit させる．ただし，

$$I = 0.000664, \quad c = 0.0000696$$

は定数とし，fitting のパラメータは， A, B, k の 3 つである．これらのパラメータの初期値は，

$$A = 0.1, \quad B = -0.1, \quad k = 0.0004$$

とする．以上のことは，gnuplot のコマンドでは，

```
f(x) = A*exp((-k+sqrt(k**2-4*I*c))/(2*I)*x) + B*exp((-k-sqrt(k**2-4*I*c))/(2*I)*x)
I = 0.000664
c = 0.0000696
A = 0.1
B = -0.1
k = 0.0004
fit f(x) "gly4.dat" using 1:3 via A,B,k
```

と入力すればよい．これを実行すれば計算が始まり，

Final set of parameters	Asymptotic Standard Error
=====	=====
A = 0.356337	+/- 0.04714 (13.23%)
B = -0.205801	+/- 0.04717 (22.92%)
k = 0.000441202	+/- 3.662e-06 (0.8301%)

という結果を得る。従って、この系の抵抗係数は、

$$k = (4.41 \pm 0.04) \times 10^{-4} \text{ kg} \cdot \text{m}^2 \cdot \text{s}^{-1}$$

と求まる（無論、この数値が妥当かどうかの議論は必要である→どのように議論すればよいか考えてみてください）。また、この運動を表すグラフは、図2である。




図 2: グリセリン水溶液中での円筒の運動（過制動）。

4 おわりに

ここまでで、gnuplot の（2次元プロットの）基本的な操作の説明は、おしまいである。「基本的な」と書いたのは、この説明が全てではないし、最善なものでもない。後は各自で、より見やすく、より分かりやすいグラフを描画するために、様々な工夫をしてほしい。最後に、gnuplot のコマンド等で困ったときに、筆者がしばしば参考にするサイトを紹介する。

- gnuplot コマンド集 (<http://www.gnuplot-cmd.com>)
- gnuplot スクリプトの解説－米澤進吾のホームページ
(<https://ss.scphys.kyoto-u.ac.jp/person/yonezawa/contents/program/gnuplot/index.html>)

また、gnuplot 以外にもグラフを描画したり、解析をしたりすることのできるソフトウェアは存在する。例えば、Mathematica や MATLAB である。なんでもいいから、何か新しい、コマンドで操作するタイプのソフトウェアの操作を身に付ける、ということは、大学生の間にやっておくべきことじゃないかなあ。時間もかかるので、なるべく早いうちから。